Project Report:

Small Hardware Development and Prototyping Board for the SX28

Project Number:

PR57

1. Project Description

2. Schematic Diagram

3. Physical Diagram

4. Component Layout Diagram

5. Bill of Materials

6. Code Listing

Submitted By:

Peter Van der Zee

Project Description

The purpose of this product is to provide a multi-function development platform for the SX28 processor. In order to facilitate this, a small printed circuit board was designed with numerous standard features that lend themselves well to the Virtual Peripheral concept. To this end, only the most economic basic circuitry is provided, relying on the programmer's ingenuity to provide fast response software to effect the operation of real world peripherals.

Although Virtual Peripheral software is not specifically part of this project, a simple task scheduler as part of the author's standard project launch point is included. Furthermore, a simple dual pulse density modulation sine-wave generator is included to demonstrate the effectiveness of the circuitry and the scheduler.

The development board measures about four inches square, and is powered by a switch mode 5 volt regulator accepting 9 to 24 V AC/DC input from a wall-wart. The unit is equipped with a socket for an SX28 processor, and two connectors each provide access to all 20 port bits for further connection to other boards. One of the main conveniences is that each of the twenty port bits is also permanently connected to a CMOS driver that in turn drives an LED. In this manner all input and output port bit states are continuously displayed.

The unit provides for communication by means of a 9 pin RS232 level port as well as an RS485 port connected to a 3 pin header as well as a RJ11 telephone jack. Both channels require software bit-banging from port A.

A serial EERAM permits power-down storage capability.

Eight input push buttons permit convenient user inputs, four each on port A and port B. A piezo element speaker permits some sound functionality.

An analog section connected to port C provides four pulse density modulated analog outputs (DAC) that are also selectably wired to accommodate charge balance analog to digital conversion in software. The processor's comparator inputs can also be connected to two of these DACs or to two single turn voltage set point pots.

Various functional conveniences are provided. One permits isolation of all of the analog section from port C on a bit-by-bit basis, Also, the comparator inputs can be totally isolated to prevent leakage from its two LED drivers. Insufficient port bits exist to permit all of the features to be available at all times, so a flexible plug selection scheme is employed to select those non conflicting functions required for the current development task.

The standard 4 pin programming connector accepts the SX Key for programming and debugging operations. A 3 pin socket permits convenient removal of the resonator while debugging, and a reset button pulls MCLR to ground.

Schematics


Title
Universal Micro Controller
UMC-11712
Size: A3
Number: 16818
Revision: 4.0
Date: 28-Feb-2005
Drawn By: C.M.
Sheet 1 of 1
This drawing contains proprietary information and may not be disclosed or copied without the express written consent of the owner, Datek Industries Inc. Edmonton, Alberta, Canada.
Used On: D1066
DATEK INDUSTRIES INC.
Research and Development
Bus Interface
Power in 9V to 24V
RS232 Port
DCE
RS485 ports
MAX232A
74HC540
LTC487/75176



REV	DESCRIPTION	DATE	BY

SCALE	DATE
DRAWN BY T.K.	05.02.28
CHECKED	
APPROVED	
USED ON D1086	
PHYSICAL DIAGRAM	

DATEK INDUSTRIES INC.
EDMONTON ALBERTA

TITLE Universal Micro Controller UMC-11712

DWG. NO.	SHEET	OF	REV
16819	1	1	



THIS DRAWING CONTAINS PROPRIETARY INFORMATION, AND MAY NOT BE DISCLOSED OR COPIED WITHOUT THE EXPRESS WRITTEN CONSENT OF THE OWNER, DATEK INDUSTRIES INC.				
REV	DESCRIPTION	DATE	BY	

SCALE	DATE
DRAWN BY T.K.	05.02.28
CHECKED	
APPROVED	
USED ON 01086	
COMPONENT DIAGRAM	

DATEK INDUSTRIES INC.
EDMONTON ALBERTA

TITLE Universal Micro Controller
UMC-11712

DWG. NO.	SHEET	OF	REV
16820	1	1	

Parts List

Except where noted, equivalent parts may be substituted.

Card UMC-11712, Rev 0, Feb 28/2005, Universal Micro Controller

Option 0 (SCL0100): No Selected components.

Item	Qty.	Designator(s)	DPN	Manufacturer	Part #	Package	Description	Comments
1	1.	Drawing #16820	13796.	AP Circuits	PCB UMC-11712	n/a	2 Sided, 1 oz, 2 Solder Masks, 1 Overlay, PTH PCB	
2	1.	U8	8176.	Microchip	24LC16B/CN	SOIC-8	Serial EEPROM, 16K, Low Power, CMOS	
3	1.	U5	2162.	Texas Instruments	SN74HC04N	PDIP-14	Hex Inverter	
4	2.	U3, U4	2164.	Texas Instruments	SN74HC540N	PDIP-20	Octal 3-State Inverting Buffer/Line Driver/Line Receiver	
5	1.	U1	13304.	Ubicom (Scenix)	SX28AC75/DP	PDIP-28(0.3")	Microcontroller, 75MHz, 8-Bit CMOS, High-Performance with EE/Flash Program Memory	(On socket). Program with ...
6	1.	U7	13012.	Linear Technology	LTC1487CN8	PDIP-8	Ultra-Low Power RS485 Transceiver with Low EMI, Shutdown and High Input Impedance	Or 75176. (On socket)
7	1.	U6	3971.21	Intersil	ICL232CPE	PDIP-16	+5V Powered RS-232 Driver/Receiver, 2 Drivers, 2 Receivers	
8	1.	U2	13144.	National	LM2574N-5.0	PDIP-8	0.5A Step-Down Switching Regulator, 50kHz	
9	1.	B1	8163.	General Instrument	DF04M	PDIP-6(with 2 middle pins missing)	Bridge Rectifier, Miniature, Glass Passivated, Single-Phase, Silicon 400V 1A	
10	1.	D21	3740.	Motorola	1N5819	59-04	Schottky Diode, 1A 40V, 0.6Vf	
11	20.	D1, D2, D3, D3, D4, D5, D6, D7, D8, D9, D10, D11, D12, D13, D14, D15, D16, D17, D18, D19, D20.	1853.	Lite-On	LTL-4221N	T-1, 0.1" pitch	T-1 Red Diffused LED, 3mm dia, (small shoulder 3.2mm)	Use LEDs from Southern Tel. stock.
12	1.	R27	2758.	Sanyo OHM	RD 1/4WP 1K2J	Axial 6x2.3mm	1K2 ±5% 1/4W 300V Carbon Film Resistor	Need 1k2 1/8W
13	4.	R2, R3, R4, R5.	2762.	Sanyo OHM	RD 1/4WP 10KJ	Axial 6x2.3mm	10K ±5% 1/4W 300V Carbon Film Resistor	Stand up on board vertically.
14	4.	R6, R7, R8, R9.	2763.	Sanyo OHM	RD 1/4WP 100KJ	Axial 6x2.3mm	100K ±5% 1/4W 300V Carbon Film Resistor	Stand up on board vertically.
15	2.	R18, R19	2906.	Sanyo OHM	RD1/8WP 100RJ	Axial 3.2x1.9mm	100R ±5% 1/8W 200V Metal Film Resistor	
16	8.	R10, R11, R12, R13, R14, R15, R16, R17.	2875.	Sanyo OHM	RD1/8WP 1KJ	Axial 3.2x1.9mm	1K ±5% 1/8W 200V Metal Film Resistor	
17	1.	R23	2885.	Sanyo OHM	RD1/8WP 2K2J	Axial 3.2x1.9mm	2K2 ±5% 1/8W 200V Metal Film Resistor	
18	2.	R29, R30	2896.	Sanyo OHM	RD1/8WP 4K7J	Axial 3.2x1.9mm	4K7 ±5% 1/8W 200V Metal Film Resistor	
19	2.	R26, R28	2623.	Sanyo OHM	RD1/8WP 7K5J	Axial 3.2x1.9mm	7K5 ±5% 1/8W 200V Metal Film Resistor	
20	3.	R1, R22, R24	2878.	Sanyo OHM	RD1/8WP 10KJ	Axial 3.2x1.9mm	10K ±5% 1/8W 200V Metal Film Resistor	
21	1.	R25	2888.	Sanyo OHM	RD1/8WP 22KJ	Axial 3.2x1.9mm	22K ±5% 1/8W 200V Metal Film Resistor	
22	2.	RP1, RP2	3188.	Bourns	4610X-101-472	SIP-10	Resistor Network, 4K7, Bussed, 10 Pin SIP	
23	2.	R20, R21	13010.	Bourns	3386W-1-102	n/a	Potentiometer, 1K, 3/8" Square, Single-Turn, Cermet, Industrial, Sealed, Side Adjust	
24	2.	C8, C10	8518.	Philips	K103K15X7RF53L2	Radial 0.1"	10nF ±10%, 50V, X7R Ceramic Capacitor, 0.1" L.S.	
25	2.	C7, C9	8507.	Philips	K104K15X7RF53L2	Radial 0.1"	100nF ±10%, 50V, X7R Ceramic Capacitor, 0.1" L.S.	
26	7.	C11, C12, C15, C16, C17, C18, C19.	1119.	AVX	SR215C104MAA	Radial 0.2"	100nF ±20%, 50V, X7R Ceramic Capacitor	
27	4.	C3, C4, C5, C6	8799.	ITT	TAP1M25SP	Radial 0.1"	1µF, 25V, ±20%, Tantalum Capacitor, 0.1" L.S.	Bend leads to 0.2".
28	2.	C2, C13	1223.	ITT	TAP22M16	Radial 0.2"	22µF, 16V, ±20%, Tantalum Capacitor, 0.2" Lead Spacing	
29	1.	C20	1064.	Marcon	CEUSM1E100	5x11mm, Radial 2mm Lead Pitch	10µF, 25V, Radial Leaded Electrolytic Capacitor, 2mm L.S	
30	1.	C14	1084.	Marcon	CEUSM1V220	6.3x11mm, Radial 2.5mm Lead Pitch	22µF, 35V, Radial Leaded Electrolytic Capacitor, 2.5mm L.S	
31	1.	C1	1088.	United Chemi-Con	KME25VB221M8X1	8x11mm, Radial 3.5mm Lead Pitch	220µF, 25V, Radial Leaded Electrolytic Capacitor, -40°C to +105°C Temperature Rating, 3.5mm L.S	

Parts List

Except where noted, equivalent parts may be substituted.

Card UMC-11712, Rev 0, Feb 28/2005, Universal Micro Controller

Option 0 (SCL0100): No Selected components.

Item	Qty.	Designator(s)	DPN	Manufacturer	Part #	Package	Description	Comments
32	1.	BP1	13550.	muRata ERiE	PKM13EPY-4002	n/a	Encased Piezo Alarm, 4 kHz 70dB@10cm@3V	
33	1.	L1	8223.	Head Strong Industries Ltd.	1mH	0.25" (dia) x 0.34" (H) Radial Leaded 0.125" Spacing	Inductor, 1mH (Choke Coil)	
34	1.	U7	4599.	Robinson Nugent	ICT-083-S-TG	n/a	DIP Socket, 8 Pin, 0.3" Wide, Solder Tail, Machined Pin	
35	1.	U1	8706.	Robinson Nugent	ICE-283-STG	n/a	DIP Socket, 28 Pin, 0.3" Wide, Solder Tail, Machined Pin	
36	.15	P9	8742.	Samtec	SS-120-T-2	n/a	Socket Strip, Single Row, 20 Pin, Tin Shell	Cut to 1x3 pins
37	.2	P7, P8, P10	1393.	AMP	4-103185-0	n/a	Breakaway Header, Unshrouded, Single Row, 0.100" Centers, 0.025" Square Straight Posts, 0.230" Tops, 0.120" Bottoms, 40 Position	Cut to 1x2 pins for P10 and 1x3 pins for P7 and P8.
38	.95	J1, J2, J5, P6	1402.	AMP	4-103186-0	n/a	Breakaway Header, Unshrouded, Double Row, 0.100"x0.100" Centers, 0.025" Square Straight Posts, 0.230" Tops, 0.120" Bottoms, 80 Position	Cut to 2x2 pins for J5, 2x8 pins for J2 and P6 and 2x20 pins for J1.
39	.075	J3	1398.	AMP	4-102974-0	n/a	Breakaway Header, Unshrouded, Single Row, 0.100" Centers, 0.025" Square Right-Angle Posts, 0.318" Tops, 0.110" Bottoms, 40 Position	Cut to 1x3 pins
40	1.	P1	13797.	Samtec	SSW-126-02-G-S-R	n/a	Socket Strip, 0.025" SQ, Solder Tail, 0.100" Centers, 26 Pins Per Row, Standard Insertion Force, Gold Contacts, Single Row, Right Angle	Cut to 1x24 pins socket. Can use DPN 8287 and bend.
41	1.	P2	8287.	Samtec	SSW-126-21-G-S	n/a	Socket Strip, 0.025" SQ, Solder Tail Pin, 0.100" Centers, 26 Pins Per Row, Low Insertion Force, Gold Contacts, Single Row, Straight	Cut to 1x24 pins socket. Can use DPN 8287.
42	1.	P5	1462.	AMP	640457-4	n/a	MTA-100 Post Header, 0.100" Centers, 4 Position, Right Angle 0.025" Square Post, Friction Lock Header with Polarizing Notches	Can use DPN 9834 and cut to length.
43	1.	P3	5978.	Taicon	DBL-09S-318	n/a	DE-9S, D-Sub, Size E, 9 Position, Sockets, Gold, Right-Angle PC Mount, 0.318" Panel to First Pin Row, Metal Shell, Fork Lock Board Mounting, Removable Hex Screwlock Face Mounting Connector	
44	1.	P4	8866.	Thomas & Betts	020.000.186 (Virginia Plastics)	n/a	Modular Female Connector, PC Mount, Right Angle, Low Profile, DT Series, 6 Position, 6 Contact	
45	1.	J6	4696.	Keystone	1266	n/a	PC Quick-Fit, 0.250" Wide, 0.312" Deep, Right Angle, Male Terminal	
46	1.	J4	12944.	CUI Stack	PJ-002A (Digikey CP-002A-ND)	n/a	Power Receptacle Connector, 2.1 mm Center Pin, PCB Mount RA	
47	10.		1503.	AMP	531220-2	n/a	2 Position, Low Profile, 0.100" Centerline with Slot, Post Shunts, "Midgie"	Qty. to be confirmed
48	9.	S1, S2, S3, S4, S5, S6, S7, S8, S9.	7256.	ITT Schadow	KSA0M211	DIP-4 pin	Key Switch, Ultra Miniature, Standard Actuator, PCB DIP	Sub. DPN 13076

```
;==============================================================================
;TITLE:           Sines.src
;
;PURPOSE:   Demonstrate a the effectiveness of an SX development board by
;           implementing a simple non-preemptive multi tasking scheduler
;           operating a dual pulse density modulation sine wave generator.
;
;AUTHOR:    Peter Van der Zee, Datek Industries Inc.
;
;REVISIONS: Feb 27, 2005 Original.
;
;CONNECTIONS:     Port: b.0   button to lower frequency 1
;                 b.1   button to raise frequency 1
;                 b.2   button to lower frequency 2
;                 b.3   button to raise frequency 2
;
;           Port: c.0   output as frequency 1  PWM output to RC filter 1.
;                 c.2   output as frequency 2  PWM output to RC filter 2.
;
;DETAILS:   Each of two independent simple sine wave generators operate by
;           pulse density modulating an output bit in a deterministic
;           Interrupt Service Routine. A tick based task scheduler controls
;           frequency selection control and sine value calculation for each
;           of the generators.
;
;           The scheduler demonstrates multiple independent tasks operating
;           without much concern of each other with   the exception of being
;           non-preemptive in nature. In other words, a task that requires
;           more rapid response will not interrupt a slower task already
;           running or scheduled to run. For greater determinism it is
;           important that no task "hogs" a lot of processor time in any
;           run instance, and it is absolutely crucial that no task uses
;           long delay loops. The purpose of the scheduler is to remove the
;           in-line     requirement for delays by letting the scheduler provide
;           those instead.
;
;           In the generators, the sine value resolution is purposely left
;           coarse so on an oscilloscope the user can see the fixed effect
;           of frequency adjustment through raise/lower buttons.
;           Finer resolution can be conveniently made by expanding access
;           and granularity of the sine lookup table, albeit at the expense
;           of maximum frequency.
;
;           It should be obvious that replacing the sine lookup table with
;           a ramp value table, a sawtooth table or   any random function table
;           that other functions can be equally easily generated.
;
;           The scheduler time ticks are set to convenient numbers, in this
;           case permitting task threads to be executed at even decades of
;           time from the base tick of 1 usec for the ISR, to 10 and 100 usec,
;           1, 10 and 100 msec, and 1 sec. The scheduler can be easily altered
;           for more or less resolution, the major stipulation being that each
;           slower tick is an integer multiple of the previous tick.
;           More complicated arrangements can of course be made. Where mutiple
;           tick (non-decade) delays are required in a thread, then the thread
;           itself is tasked with the requirement to do so.
;==============================================================================
```

```
;---------------DEVICE DIRECTIVES---------------------------------------------

id          'Sines'

            DEVICE      SX28,oschs3,stackx,turbo

            FREQ  50_000_000  ;default run speed = 50MHz
            RESET ResetEntry  ;jump to start label on reset

;---------------CONSTANTS-----------------------------------------------------

Dac1Bit           equu   rc.0  ;pulse density modulator 1 output to RC integrator
Dac2Bit           equ   rc.2  ;pulse density modulator 2 output to RC integrator
IntValue    equ   -50         ;interrupt reload value for 1 micro sec at 50 MHz
Ram1        equ   $10         ;

;---------------VARIABLES-----------------------------------------------------

            org   8
Flags       ds    1
Intflag     equ   Flags.0     ;interrupt occurred flag

            org   Ram1
Timer10uS   ds    1           ;counter to get to 10uSec
Timer100uS  ds    1           ;counter to get to 100uSec
Timer1mS    ds    1           ;counter to get to 1mSec
Timer10mS   ds    1           ;counter to get to 10mSec
Timer100mS  ds    1           ;counter to get to 100mSec
Timer1S     ds    1           ;counter to get to 1Sec

Dac1Value   ds    1           ;value for the PWM 1 output
Dac1Accum   ds    1           ;accumulator for PWM 1
Period1     ds    1           ;duration of one cycle of frequency 1
Period1Load ds    1           ;duration of one cycle load source for frequency 1
F1index     ds    1           ;index into sine table for frequency 1
Dac2Value   ds    1           ;value for the PWM 2 output
Dac2Accum   ds    1           ;accumulator for PWM 2
Period2     ds    1           ;duration of one cycle of frequency 2
Period2Load ds    1           ;duration of one cycle load source for frequency 2
F2index     ds    1           ;index into sine table for frequency 2
```

```
;---------------INTERRUPT ROUTINE---------------------------------------------

            org   0
Intsvc
;For each of two one byte PWMs, calculate the rollover carry and then clear or
;set the PWM bit accordingly
;The add-with-carry option must be disabled unless carry is specifically cleared
;before the add.

            setb  Intflag     ;advise scheduler an interrupt has occurred
            add   Dac1Accum,Dac1Value     ;calculate PWM 1 overflow
            sc                ;
            clrb  Dac1Bit     ;clear PWM 1
            snc               ;
            setb  Dac1Bit     ;set PWM 1
            add   Dac2Accum,Dac2Value     ;calculate PWM 2 overflow
            sc                ;
            clrb  rc.2        ;clear PWM 2
            snc               ;
            setb  rc.2        ;set PWM 2
            mov   w,#IntValue ;
            retiw             ;return from interrupt and reset for 50 instr

;---------------INITIALIZATION------------------------------------------------

ResetEntry
            ;Initialize the ports
SetLevels   mov   m,#$0d            ;Set 0 for CMOS levels
            mov   !ra,#%0000        ;
            mov   !rb,#%0000_0000   ;
            mov   !rc,#%0000_0000   ;
SetPullups  mov   m,#$0e            ;Set 0 for pullups
            mov   !ra,#%0000        ;port a not used
            mov   !rb,#%0000_0000   ;input buttons
            mov   !rc,#%1111_1111   ;
SetTris     mov   m,#$0f            ;Set 0 for output
            clr   ra                ;
            mov   !ra,#%1111        ;port a not used
            clr   rb                ;
            mov   !rb,#%0000_1111   ;X,X,X,X _ F2up,F2dn,F1up,F1dn
            clr   rc                ;
            mov   !rc,#%0000_0000   ;X,X,X,X _ X,DAC2,X,DAC1

            ;Clear memory
Clearmem    mov   fsr,#$10          ;point to first memory bank
Clearone    setb  fsr.4             ;stay in proper half
            clr   ind               ;clear this location
            incsz fsr               ;point to next location
            jmp   Clearone          ;not at end so clear one more

            ;Initialize the scheduler timers
            mov   w,#10             ;timer decade value
            mov   Timer10uS,w       ;10 microseconds
            mov   Timer100uS,w      ;100 microseconds
            mov   Timer1mS,w        ;1 millisecond
            mov   Timer10mS,w       ;10 milliseconds
            mov   Timer100mS,w      ;100 milliseconds
            mov   Timer1S,w         ;1 second
```

```
            ;Initialize the variables
            clr   rtcc              ;
            mov   !option,#%1000_1000 ;internal rtcc
            clr   Flags             ;
            mov   Dac1Value,#128    ;set initial value of dac1 half way
            mov   Dac2Value,#128    ;set initial value of dac2 half way

;---------------MAIN PROGRAM-------------------------------------------------

Main
;The scheduler keeps time for the whole system and triggers sine calculations
;for both generators each 10 microseconds.
;Every 100 milliseconds it looks for raise/lower buttons being pushed, and if
;so,calls the corresponding generator's raise/lower routine.

            sb    Intflag           ;test for interrupt occurred
            jmp   Main              ;wait for interrupt
            bank  Ram1              ;
Usec1       clrb  Intflag           ;clear that fact
            decsz Timer10uS         ;scheduler 1 usec base tick
            jmp   Main              ;wait for occurrence of next interrupt
Usec10      mov   Timer10uS,#10           ;reload 10usec timer
            call  Sine1             ;determine freq 1 step
            call  Sine2             ;determine freq 2 step
            decsz Timer100uS        ;scheduler 10 usec tick
            jmp   Main              ;wait for occurrence of next interrupt
Usec100     mov   Timer100uS,#10    ;reload 10usec timer
            ;put 100 uSec routines here
            decsz Timer1mS          ;scheduler 100 usec tick
            jmp   Main              ;wait for occurrence of next interrupt
Msec1       mov   Timer1mS,#10      ;reload 100usec timer
            ;put 1 mSec routines here
            decsz Timer10mS         ;scheduler 1 msec tick
            jmp   Main              ;wait for occurrence of next interrupt
Msec10      mov   Timer10mS,#10     ;reload 1msec timer
            ;put 10 mSec routines here
            decsz Timer100mS        ;scheduler 1 usec base tick
            jmp   Main              ;wait for occurrence of next interrupt
Msec100     mov   Timer100mS,#10    ;reload 10usec timer
            sb    rb.0              ;test button for lower frequency 1
            call  Lower1            ;decrease frequency 1
            sb    rb.1              ;test button for higher frequency 1
            call  Higher1           ;increase frequency 1
            sb    rb.2              ;test ▯button ▯for lower frequency 2
            call  Lower2            ;decrease frequency 2
            sb    rb.3              ;test button for higher frequency 2
            call  Higher2           ;increase frequency 2
            decsz Timer1S           ;scheduler 1 usec base tick
            jmp   Main              ;wait for occurrence of next interrupt
Sec1        mov   Timer1S,#10       ;reload 10usec timer
            ;put 1 Sec routines here
            jmp   Main              ;wait for occurrence of next interrupt
```

```
;---------------SUBROUTINES-----------------------------------------------------

Lower1      ;reduce frequency of generator 1 but not below zero
            incsz Period1Load       ;increase the period of frequency 1
            skip                    ;
            dec   Period1Load       ;underflow not permitted
            retp                    ;

Higher1     ;increase frequency of generator 1 but not above $ff
            decsz Period1Load       ;decrease the period of frequency 1
            skip                    ;
            inc   Period1Load       ;overflow not permitted
            retp                    ;

Lower2      ;reduce frequency of generator 2 but not below zero
            incsz Period2Load       ;increase the period of frequency 2
            skip                    ;
            dec   Period2Load       ;underflow not permitted
            retp                    ;

Higher2     ;increase frequency of generator 2 but not above $ff
            decsz Period2Load       ;decrease the period of frequency 2
            skip                    ;
            inc   Period2Load       ;overflow not permitted
            retp                    ;

Sine1       ;calculate lookup time for generator 1, and if so, get sine value
            decsz Period1           ;step frequency 1 period duration
            retp                    ;not time for lookup; return to scheduler
            mov   Period1,Period1Load     ;reload period 1 timer
            inc   F1index           ;step to next sine value in lookup table
            mov   w,F1index         ;
            call  SineLookup        ;get sine value for this index
            mov   Dac1Value,w       ;setup new dac 1 value for the ISR
            retp                    ;done freq1; return to scheduler

Sine2       ;calculate lookup time for generator 2, and if so, get sine value
            decsz Period2           ;step frequency 2 period duration
            retp                    ;not time for lookup; return to scheduler
            mov   Period2,Period2Load     ;reload period 2 timer
            inc   F2index           ;step to next sine value in lookup table
            mov   w,F2index         ;
            call  SineLookup        ;get sine value for this index
            mov   Dac2Value,w       ;setup new dac 2 value for the ISR
            retp                    ;done freq2; return to scheduler

SineLookup  ;lookup the sine value of index in w
            and   w,#%0000_1111     ;only use 16 steps in lookup table
            add   pc,w              ;calculate offset into lookup table
Sin0        retw  128               ;$80
Sin1        retw  177               ;$b1
Sin2        retw  218               ;$da
Sin3        retw  245               ;$f5
Sin4        retw  255               ;$ff
Sin5        retw  245               ;$f5
Sin6        retw  218               ;$da
Sin7        retw  177               ;$b1
Sin8        retw  128               ;$80
Sin9        retw  79                ;$4f
```

```
SinA        retw  38                ;$26
SinB        retw  11                ;$b
SinC        retw  1                 ;$1
SinD        retw  11                ;$b
SinE        retw  38                ;$26
SinF        retw  79                ;$4f
```